NMST 434,
Exercise session VII : M –estimators II

April 8, 2019

Example: M-estimators of location for univariate distributions

Truedist is the true distribution from the location family whose parameter we estimate
Mdist is the distribution whose maximum likelihood estimating equations we use for the construction of M estimators. In this case, function $\psi 0$ is taken to be $-\partial \mathrm{Log}[f[x;\ \mu]]/\partial\ \mu$.
alternatively, function $\psi 0$ can be specified directly as in the case of the Huber loss function or the Tukey biweight function.
In the special case *Mdist=Truedist* we get the usual maximum likelihood estimators as the M-estimators, and the asymptotic variance in the final computation is, in fact, the Rao-Cramér bound.

```
Clear["Global`*"]
$Assumptions = Element[μ, Reals] && b > 0;
(*Mdist=NormalDistribution[];*)
Mdist = StudentTDistribution[1];
Truedist  = NormalDistribution[];
(*Truedist=LogisticDistribution[];*)
(*Truedist=ExponentialDistribution[1];*)
(*Truedist=StudentTDistribution[6];*)
f[x_] = PDF[Mdist, x];
ψ0[t_] = - D[f[t], t]/f[t];
(*b=1;*)
ψ0[t_] = Min[b, Max[t, -b]];(* !!! For non-differentiable functions ψ0,
the asymptotic variance matrices may be incorrect - formally,
one would have to use Theorem 10 !!! *)
(*ψ0[t_]=t(1-t²/b²)²Boole[Abs[t]≤b];*)
(*Plot[ψ0[t],{t,-3b/2,3b/2},PlotLabel→"Function ψ"]*)
ψ[x_] = ψ0[x - μ];
Γ = Integrate[(D[ψ[x], {μ}] /. μ → 0) PDF[Truedist, x], {x, -∞, ∞}]
Σ = Integrate[(ψ[x]² /. μ → 0) PDF[Truedist, x], {x, -∞, ∞}]
Var1 = Σ / (Γ²)
```

$$-\mathrm{Erf}\left[\frac{b}{\sqrt{2}}\right]$$

$$e^{-\frac{b^2}{2}}\left(-b\sqrt{\frac{2}{\pi}}+e^{\frac{b^2}{2}}\,\mathrm{Erf}\left[\frac{b}{\sqrt{2}}\right]+b^2\,e^{\frac{b^2}{2}}\,\mathrm{Erfc}\left[\frac{b}{\sqrt{2}}\right]\right)$$

$$\frac{e^{-\frac{b^2}{2}}\left(-b\sqrt{\frac{2}{\pi}}+e^{\frac{b^2}{2}}\,\mathrm{Erf}\left[\frac{b}{\sqrt{2}}\right]+b^2\,e^{\frac{b^2}{2}}\,\mathrm{Erfc}\left[\frac{b}{\sqrt{2}}\right]\right)}{\mathrm{Erf}\left[\frac{b}{\sqrt{2}}\right]^2}$$

For simple distributions (normal or exponential) we plot the mean, median, and the estimating function of the M-estimator (as a function of the tuning parameter b). The root of the estimating function is the M-estimate.

```
OF = Integrate[(ψ[x]) PDF[Truedist, x], {x, -∞, ∞}];
Manipulate[Plot[OF /. b → B, {μ, -1, 2},
   PlotLabel → "Identifying function", Epilog → {Directive[Red, Thick],
     Line[{{Mean[Truedist], -5}, {Mean[Truedist], 5}}], Directive[Orange, Thick],
     Line[{{Median[Truedist], -5}, {Median[Truedist], 5}}]}], {{B, 1}, 0, 5}]
Mean[Truedist]
Median[Truedist]
```




```
0
```

```
0
```

Computation of the asymptotic variance of Huber's M-estimator of location using Theorem 10, see Example 40.

```
Var2 =
  Integrate[x² PDF[Truedist, x], {x, -b, b}] + b² (1 - CDF[Truedist, b] + CDF[Truedist, -b])
  ------------------------------------------------------------------------------------------
                         (CDF[Truedist, b] - CDF[Truedist, -b])²
% //
 N
```

$$\frac{-b\, e^{-\frac{b^2}{2}} \sqrt{\frac{2}{\pi}} + \mathrm{Erf}\left[\frac{b}{\sqrt{2}}\right] + b^2 \left(1 - \frac{1}{2}\mathrm{Erfc}\left[-\frac{b}{\sqrt{2}}\right] + \frac{1}{2}\mathrm{Erfc}\left[\frac{b}{\sqrt{2}}\right]\right)}{\left(\frac{1}{2}\mathrm{Erfc}\left[-\frac{b}{\sqrt{2}}\right] - \frac{1}{2}\mathrm{Erfc}\left[\frac{b}{\sqrt{2}}\right]\right)^2}$$

```
(-0.797885 × 2.71828^(-0.5 b²) b + Erf[0.707107 b] +
   b² (1. - 0.5 Erfc[-0.707107 b] + 0.5 Erfc[0.707107 b])) /
 (0.5 Erfc[-0.707107 b] - 0.5 Erfc[0.707107 b])²
```

Comparison of the asymptotic variances for Huber's estimator computed using Theorems 9 and 10. Neither variance expression is true if the distribution is not symmetric, i.e. when μ is misspecified.

```
Plot[{Var1, Var2}, {b, 0, 5}, PlotRange → All, PlotStyle → {{Thick}, Dashed},
 PlotLabel → "AVar of Huber's estimator as a function of the parameter b"]
```



Example: M-estimation of location and scale for univariate distributions

Truedist is the true distribution from the location-scale family whose parameters we estimate
Mdist is the distribution whose maximum likelihood estimating equations we use for the construction of M estimators. In this case, function ψ0 is taken to be $-\partial \mathrm{Log}[f[x;\mu,\sigma]]/\partial\mu$.
alternatively, function ψ0 can be specified directly as in the case of the Huber loss function or the Tukey biweight function.
In the special case *Mdist=Truedist* we get the usual maximum likelihood estimators as the M-estimators, and the asymptotic variance in the final computation is, in fact, the Rao-Cramér bound.

```
Clear["Global`*"]
$Assumptions = Element[μ, Reals] && σ > 0;
(*Mdist=NormalDistribution[];*)
Mdist = StudentTDistribution[1];
Truedist  = NormalDistribution[];(*LogisticDistribution[];*)
(*Truedist=StudentTDistribution[6];*)
f[x_] = PDF[Mdist, x];
ψ0[t_] = - D[f[t], t] / f[t];
b = 1;
(*ψ0[t_]=Min[b,Max[t,-b]];(* !!! For non-differentiable functions ψ0,
the asymptotic variance matrices may be incorrect - formally,
one would have to use Theorem 10 !!! *) *)
(*ψ0[t_]=t (1-t^2/b^2)^2 Boole[Abs[t]≤b];*)
Plot[ψ0[t], {t, -3 b / 2, 3 b / 2}, PlotLabel → "Function ψ"]
κ = Integrate[ψ0[x] x PDF[Truedist, x], {x, -∞, ∞}];
(* standardization constant that ensures correct identification of parameter σ *)
N[κ]
ψ[x_] = {ψ0[(x - μ)/σ], ψ0[(x - μ)/σ] (x - μ)/σ - κ};
D[ψ[x], {{μ, σ}}]
Outer[Times, ψ[x], ψ[x]]
```



0.688641

$$\left\{\left\{\frac{4\,(x-\mu)^2}{\left(1+\frac{(x-\mu)^2}{\sigma^2}\right)^2\sigma^3}-\frac{2}{\left(1+\frac{(x-\mu)^2}{\sigma^2}\right)\sigma},\ \frac{4\,(x-\mu)^3}{\left(1+\frac{(x-\mu)^2}{\sigma^2}\right)^2\sigma^4}-\frac{2\,(x-\mu)}{\left(1+\frac{(x-\mu)^2}{\sigma^2}\right)\sigma^2}\right\},\right.$$
$$\left.\left\{\frac{4\,(x-\mu)^3}{\left(1+\frac{(x-\mu)^2}{\sigma^2}\right)^2\sigma^4}-\frac{4\,(x-\mu)}{\left(1+\frac{(x-\mu)^2}{\sigma^2}\right)\sigma^2},\ \frac{4\,(x-\mu)^4}{\left(1+\frac{(x-\mu)^2}{\sigma^2}\right)^2\sigma^5}-\frac{4\,(x-\mu)^2}{\left(1+\frac{(x-\mu)^2}{\sigma^2}\right)\sigma^3}\right\}\right\}$$

$$\left\{\left\{\frac{4\,(x-\mu)^2}{\left(1+\frac{(x-\mu)^2}{\sigma^2}\right)^2\sigma^2},\ \frac{2\,(x-\mu)\left(-2+\frac{2\,(x-\mu)^2}{\left(1+\frac{(x-\mu)^2}{\sigma^2}\right)\sigma^2}+\sqrt{2\,e\,\pi}\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right]\right)}{\left(1+\frac{(x-\mu)^2}{\sigma^2}\right)\sigma}\right\},\right.$$
$$\left\{\frac{2\,(x-\mu)\left(-2+\frac{2\,(x-\mu)^2}{\left(1+\frac{(x-\mu)^2}{\sigma^2}\right)\sigma^2}+\sqrt{2\,e\,\pi}\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right]\right)}{\left(1+\frac{(x-\mu)^2}{\sigma^2}\right)\sigma},\right.$$
$$\left.\left.\left(-2+\frac{2\,(x-\mu)^2}{\left(1+\frac{(x-\mu)^2}{\sigma^2}\right)\sigma^2}+\sqrt{2\,e\,\pi}\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right]\right)^2\right\}\right\}$$

Check that the estimator of scale is (Fisher) consistent, i.e. that it identifies σ

```
ϵ = 1/1000;
Integrate[(ψ0[(x - μ)/σ] (x - μ)/σ - κ /. μ → 0 // Simplify) PDF[Truedist, x], {x, -∞, ∞}]
Plot[%, {σ, ϵ, Max[b, 2]}]
FindRoot[%%, {σ, ϵ, 2}]
```

$$\sqrt{2 e \pi}\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right] - e^{\frac{\sigma^2}{2}} \sqrt{2\pi}\ \sigma\, \mathrm{Erfc}\left[\frac{\sigma}{\sqrt{2}}\right]$$



{σ → 1.}

Matrices Γ and Σ for the computation of influence functions and asymptotic variances

```
Γ = Integrate[(D[ψ[x], {{μ, σ}}] /. {μ → 0, σ → 1}) PDF[Truedist, x], {x, -∞, ∞}]
Σ = Integrate[(Outer[Times, ψ[x], ψ[x]] /. {μ → 0, σ → 1}) PDF[Truedist, x], {x, -∞, ∞}]
```

$$\left\{\left\{-2 + \sqrt{2 e \pi}\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right], 0\right\}, \left\{0, 2 - 2\sqrt{2 e \pi}\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right]\right\}\right\}$$

$$\left\{\left\{-2 + 2\sqrt{2 e \pi}\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right], 0\right\}, \left\{0, 2 - 2 e \pi\, \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right]^2\right\}\right\}$$

Influence functions of the M-estimators of location and scale

```
Plot[Evaluate[-Inverse[Γ].ψ[x] /. {μ → 0, σ → 1} // Simplify], {x, -3 b/2, 3 b/2},
 PlotLabel → "Influence functions of the M-estimators of location and scale",
 PlotLegends → {"Location", "Scale"}]
Limit[ψ[t], t → ∞]
```



$$\left\{0,\ \sqrt{2\,e\,\pi}\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right]\right\}$$

Asymptotic variance matrix of the M-estimators of location and scale

```
Inverse[Γ].Σ.Inverse[Γ] /. {μ → 0, σ → 1} // Simplify // MatrixForm
% // N // MatrixForm
```

$$\begin{pmatrix} \dfrac{2\left(-1+\sqrt{2\,e\,\pi}\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right]\right)^3}{\left(2-3\sqrt{2\,e\,\pi}\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right]+2\,e\,\pi\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right]^2\right)^2} & 0 \\ 0 & \dfrac{\left(-2+\sqrt{2\,e\,\pi}\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right]\right)^2\left(2-2\,e\,\pi\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right]^2\right)}{\left(4-6\sqrt{2\,e\,\pi}\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right]+4\,e\,\pi\ \mathrm{Erfc}\left[\frac{1}{\sqrt{2}}\right]^2\right)^2} \end{pmatrix}$$

$$\begin{pmatrix} 1.31312 & 0. \\ 0. & 0.722933 \end{pmatrix}$$

Fisher information matrix of the true distribution, to compare with the asymptotic variance of the M-estimators above

$$U = -\frac{D\left[\frac{1}{\sigma}\,\mathrm{PDF}\left[\mathrm{Truedist},\ \frac{t-\mu}{\sigma}\right],\ \{\{\mu,\ \sigma\}\}\right]}{\frac{1}{\sigma}\,\mathrm{PDF}\left[\mathrm{Truedist},\ \frac{t-\mu}{\sigma}\right]};$$

```
FIM = Integrate[Outer[Times, U, U] PDF[Truedist, t] /. {μ → 0, σ → 1}, {t, -∞, ∞}];
Inverse[FIM] // MatrixForm
% // N // MatrixForm
```

$$\begin{pmatrix} 1 & 0 \\ 0 & \frac{1}{2} \end{pmatrix}$$

$$\begin{pmatrix} 1. & 0. \\ 0. & 0.5 \end{pmatrix}$$

Derivative of log(det(Σ)) is $(\Sigma^{-1})^T$

```
d = 5; (* dimension *)
A = D[Log[Det[Table[x_{i,j}, {i, 1, d}, {j, 1, d}]]], {Table[x_{i,j}, {i, 1, d}, {j, 1, d}]}];
Short[A]
B = Transpose[Inverse[Table[x_{i,j}, {i, 1, d}, {j, 1, d}]]];
Cond = Flatten[Table[x_{i,j} → RandomReal[], {i, 1, d}, {j, 1, d}]]
(* Evaluate the matrices randomly *)
(A /. Cond)
(B /. Cond)
```

```
{{<<1>> / ((<<35>> + x_{1,2} x_{2,3} x_{3,4} x_{4,5}) x_{5,1} - (<<1>>) x_{5,2} + <<1>> -
    (<<35>> + <<1>>) <<1>> + (<<35>> + x_{1,1} x_{2,2} x_{3,3} x_{4,4}) x_{5,5}), <<1>>/<<1>>, <<1>>/<<1>>, <<1>>/<<1>>,
  ((<<1>>) x_{5,1} - (<<1>>) x_{5 <<1>> <<1>>} + <<1>> - (<<1>>) x_{5,4}) / (<<6>> + (<<35>> + <<1>>) x_{<<1>>})}, <<3>>, {<<1>>/<<1>>, <<4>>}}
```

```
{x_{1,1} → 0.366887, x_{1,2} → 0.494812, x_{1,3} → 0.240495, x_{1,4} → 0.776986, x_{1,5} → 0.41083,
 x_{2,1} → 0.700019, x_{2,2} → 0.174995, x_{2,3} → 0.000197526, x_{2,4} → 0.212031, x_{2,5} → 0.819566,
 x_{3,1} → 0.793623, x_{3,2} → 0.224809, x_{3,3} → 0.225649, x_{3,4} → 0.504024, x_{3,5} → 0.774389,
 x_{4,1} → 0.210544, x_{4,2} → 0.214303, x_{4,3} → 0.40724, x_{4,4} → 0.397694, x_{4,5} → 0.77483,
 x_{5,1} → 0.0818745, x_{5,2} → 0.675729, x_{5,3} → 0.329053, x_{5,4} → 0.898553, x_{5,5} → 0.669066}
```

```
{{12.5267, 38.8483, 24.2904, -31.0934, -10.9561},
 {5.2204, 19.1313, 8.73438, -15.6144, -3.28618},
 {-9.57831, -37.0004, -20.7786, 30.1337, 8.29059},
 {5.15826, 18.0134, 13.7601, -15.9884, -4.11902},
 {-8.97409, -25.3249, -17.5, 21.8576, 7.42188}}
```

```
{{12.5267, 38.8483, 24.2904, -31.0934, -10.9561},
 {5.2204, 19.1313, 8.73438, -15.6144, -3.28618},
 {-9.57831, -37.0004, -20.7786, 30.1337, 8.29059},
 {5.15826, 18.0134, 13.7601, -15.9884, -4.11902},
 {-8.97409, -25.3249, -17.5, 21.8576, 7.42188}}
```

Example: Multivariate M-estimation of location and scatter

```
Clear["Global`*"];
$Assumptions = μ1 ∈ Reals && μ2 ∈ Reals && σ1 > 0 && σ2 > 0 && -1 < ρ < 1;
μ = {μ1, μ2};
Σ = {{σ1^2, ρ σ1 σ2}, {ρ σ1 σ2, σ2^2}};
θ = {μ1, μ2, σ1, ρ, σ2};
Truedist = MultinormalDistribution[DiagonalMatrix[{1, 1}]];
b = 1;
(*ψ0[t_]=Min[b,Max[t,-b]]/t;(* !!! For non-differentiable functions ψ0,
the asymptotic variance matrices may be incorrect - formally,
one would have to use Theorem 10 !!! *) *)
(*ψ0[t_]=1;*)
ψ0[t_] = (2 + 1)/(t^2 + 1);
(* Corresponds to t-distribution with 1 degree of freedom - Cauchy distribution *)
(*ψ0[t_]=t(1-t^2/b^2)^2 Boole[Abs[t]≤b]/t;*)
Plot3D[Evaluate[
  ψ0[√(({x1, x2} - {0, 0}).Inverse[DiagonalMatrix[{1, 1}]].({x1, x2} - {0, 0}))]
   {x1, x2}], {x1, -3 b/2, 3 b/2}, {x2, -3 b/2, 3 b/2}, PlotRange → All,
 PlotLabel → "Functions ψ for location parameters", ImageSize → Large]
Plot3D[Evaluate[
  ψ0[√(({x1, x2} - {0, 0}).Inverse[DiagonalMatrix[{1, 1}]].({x1, x2} - {0, 0}))]
   {x1^2, x1 x2, x2^2}], {x1, -3 b/2, 3 b/2}, {x2, -3 b/2, 3 b/2}, PlotRange → All,
 PlotLabel → "Functions ψ for scatter parameters", ImageSize → Large]
κ = NIntegrate[
  ψ0[√(({x1, x2} - {0, 0}).Inverse[DiagonalMatrix[{1, 1}]].({x1, x2} - {0, 0}))]
   x1^2 PDF[Truedist, {x1, x2}], {x1, -∞, ∞}, {x2, -∞, ∞}]
ψ[x1_, x2_] = Delete[Join[ψ0[√(({x1, x2} - μ).Inverse[Σ].({x1, x2} - μ))] ({x1, x2} - μ),
     Flatten[ψ0[√(({x1, x2} - μ).Inverse[Σ].({x1, x2} - μ))] / κ
        Outer[Times, {x1, x2} - μ, {x1, x2} - μ] - Σ]], 5];
```

Functions ψ for location parameters






```
0.807817
```

Check that we correctly identified all the parameters. Integrals should be all zero at the true value of the parameters.
(addition and subtraction of 1 only to avoid warnings “Integral may be zero”)

```
NIntegrate[
  (1 + (ψ[x1, x2] /. {μ1 → 0, μ2 → 0, σ1 → 1, σ2 → 1, ρ → 0})) PDF[Truedist, {x1, x2}],
  {x1, -∞, ∞}, {x2, -∞, ∞}, AccuracyGoal → 10, WorkingPrecision → 10] - 1
```

```
{-7.1348 × 10^-6, -7.1348 × 10^-6, -8.0673 × 10^-6, -0.0000350292, -8.0673 × 10^-6}
```

If the integrals in Γ and V can be computed exactly (only for the estimating equations from the multivariate normal distribution)

```
Γ =
 Integrate[(D[ψ[x1, x2], {{μ1, μ2, σ1, ρ, σ2}}] /. {μ1 → 0, μ2 → 0, σ1 → 1, σ2 → 1, ρ → 0})
   PDF[Truedist, {x1, x2}], {x1, -∞, ∞}, {x2, -∞, ∞}]
V = Integrate[(Outer[Times, ψ[x1, x2], ψ[x1, x2]] /. {μ1 → 0, μ2 → 0, σ1 → 1,
      σ2 → 1, ρ → 0}) PDF[Truedist, {x1, x2}], {x1, -∞, ∞}, {x2, -∞, ∞}]
```

Numerical computation of matrices Γ and V

```
Γ = NIntegrate[
   (1 + D[ψ[x1, x2], {{μ1, μ2, σ1, ρ, σ2}}] /. {μ1 → 0, μ2 → 0, σ1 → 1, σ2 → 1, ρ → 0})
    PDF[Truedist, {x1, x2}], {x1, -∞, ∞},
   {x2, -∞, ∞}, AccuracyGoal → 10, WorkingPrecision → 10] - 1
V = NIntegrate[(1 + Outer[Times, ψ[x1, x2], ψ[x1, x2]] /.
      {μ1 → 0, μ2 → 0, σ1 → 1, σ2 → 1, ρ → 0}) PDF[Truedist, {x1, x2}],
   {x1, -∞, ∞}, {x2, -∞, ∞}, AccuracyGoal → 10] - 1
```

$\{\{-0.8078349055, 0.000051534, -0.0000200891, 0.000201665, 0.000201665\},$
$\{0.000051534, -0.8078349055, 0.000201665, 0.000201665, -0.0000200891\},$
$\{1.813\times10^{-6}, 0.000086699, -1.03528420564, -6.8513\times10^{-6}, 0.321575302\},$
$\{-0.0000118492, -0.0000118492, -6.8513\times10^{-6}, -0.6783948821, -6.8513\times10^{-6}\},$
$\{0.000086699, 1.813\times10^{-6}, 0.321575302, -6.8513\times10^{-6}, -1.03528420564\}\}$

$\{\{0.864823, 1.65399\times10^{-8}, -5.9907\times10^{-10}, 8.71367\times10^{-9}, 1.27134\times10^{-8}\},$
$\{1.65399\times10^{-8}, 0.864823, 1.27134\times10^{-8}, 8.71367\times10^{-9}, -5.99071\times10^{-10}\},$
$\{-5.9907\times10^{-10}, 1.27134\times10^{-8}, 0.791338, 1.71269\times10^{-8}, -0.402887\},$
$\{8.71367\times10^{-9}, 8.71367\times10^{-9}, 1.71269\times10^{-8}, 0.597113, 1.71269\times10^{-8}\},$
$\{1.27134\times10^{-8}, -5.99071\times10^{-10}, -0.402887, 1.71269\times10^{-8}, 0.791338\}\}$

Influence function of the M-estimators of location and scatter

```
IF =
  Evaluate[-Inverse[Γ].ψ[x1, x2] /. {μ1 → 0, μ2 → 0, σ1 → 1, σ2 → 1, ρ → 0} // Simplify];
Plot3D[IF, {x1, -3 b / 2, 3 b / 2}, {x2, -3 b / 2, 3 b / 2},
 PlotLabel → "Influence functions of bivariate M-estimators of location and scatter",
 PlotLegends → {"μ1", "μ2", "σ1", "ρ", "σ2"},
 AxesLabel → {"x1", "x2", "IF(x1,x2)"}, ImageSize → Large, PlotRange → All]
Plot3D[IF[[4]], {x1, -3 b / 2, 3 b / 2}, {x2, -3 b / 2, 3 b / 2},
 PlotLabel → "Influence function of the M-estimator of correlation coef.",
 PlotLegends → {"ρ"}, AxesLabel → {"x1", "x2", "IF(x1,x2)"},
 ImageSize → Large, PlotRange → All]
```





Asymptotic variance matrix of the M-estimator of location and scatter

```
Chop[Inverse[Γ].V.Inverse[Γ], 10^-10] // MatrixForm
```

$$\begin{pmatrix} 1.3252 & 0.000169101 & 0.0000569996 & 0.000717912 & 0.000451613 \\ 0.000169101 & 1.3252 & 0.000451613 & 0.000717912 & 0.0000569996 \\ 0.0000484434 & 0.000199507 & 0.705628 & -0.0000200074 & 0.0569682 \\ -0.0000421696 & -0.0000421696 & -0.000020135 & 1.29745 & -0.000020135 \\ 0.000199507 & 0.0000484434 & 0.0569682 & -0.0000200074 & 0.705628 \end{pmatrix}$$

Inverse Fisher information matrix for multivariate normal distribution

```
U = -(D[PDF[MultinormalDistribution[{μ1, μ2}, Σ], {x1, x2}], {θ}] /
      PDF[MultinormalDistribution[{μ1, μ2}, Σ], {x1, x2}]) // Simplify;
FIM = Integrate[(Outer[Times, U, U] PDF[MultinormalDistribution[
          DiagonalMatrix[{1, 1}]], {x1, x2}]) /.
     {μ1 → 0, μ2 → 0, σ1 → 1, σ2 → 1, ρ → 0}, {x1, -∞, ∞}, {x2, -∞, ∞}];
Inverse[FIM] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & \frac{1}{2} & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & \frac{1}{2} \end{pmatrix}$$